CULTURAL HIATUS AND CHRONOLOGICAL RESOLUTION: SIMULATING THE MONO CRATERS ERUPTION OF CA. A.D. 880 IN THE ARCHAEOLOGICAL RECORD

Don Laylander
California Department of Transportation
Bishop, CA 93514

ABSTRACT

One cultural response to environmental change is the temporary abandonment of a region. Under what conditions will an occupational hiatus be detectable in the archaeological record? A statistical model of obsidian hydration readings is applied to a hypothetical hiatus following the volcanic eruption of ca. A.D. 880 in the Mono Basin of east-central California. The model evaluates the importance of hiatus length, obsidian sample size, and hydration error in the archaeological visibility of such a hiatus.

Around A.D. 880, a large ("Plinean") volcanic eruption occurred in the Mono Craters just east of the Sierra Nevada. The eruption scattered deposits of ash, which have been termed "Tephra 2," up to 100 kilometers from the probable source at Panum Pumice Pit near Mono Lake. The A.D. 880 eruption is visible in the archaeological record of the region; for instance, at site CA-MNO-891, where cultural deposits both underlie and overlie a stratum identified as Tephra 2 (cf. Hall 1983; Wickstrom and Jackson 1993; Wood 1977; Wood and Brooks 1979) .

Local consequences for human populations from the eruption can be imagined. The event may have directly caused some loss of life or frightened the surviving witnesses into leaving the Mono Basin. The decimation of plant and animal communities may have drastically reduced the resource value of the affected area for humans for some time.

Two types of cultural responses to the volcanic event can be hypothesized. One type would be a long-term disjunction between pre-event and post-event conditions, such as an ethnic replacement or a permanent change in the settlement-subsistence system. The other type -- the one to be considered here -- is the less conspicuous but perhaps more likely response of an occupational hiatus, followed by a return to pre-event conditions. If such a hiatus occurred after the A.D. 880 eruption, would it be detectable archaeologically? If a hiatus is detected, could its duration be estimated archaeologically?

The key to an answer to those questions lies in the problem of chronological resolution. For the eastern Sierra region, there are three main types of usable archaeological chronologies. The first is based on artifact types -- primarily projectile points, but also shell beads, ceramics, and some other artifacts. Although this chronology is fundamental to most studies, it has a resolution of, at best, several centuries, and is clearly not suited to addressing the problem raised here. A second chronology is based on radiocarbon dating. Radiocarbon has the widest acceptance as a reliable absolute chronology, and the resolution of individual dates is fairly good. However, radiocarbon evidence has severe disadvantages for addressing the problem of identifying and measuring a hiatus, including the limited number of independent, culturally-related samples which can be obtained and the high unit cost of dating such samples. Obsidian hydration dating, with its abundance of relevant samples and fairly low unit-cost, offers the best hope for detecting a hiatus, if one existed.

PARAMETERS

Could obsidian hydration be used to detect an event such as a hiatus following the A.D. 880

eruption? Seven parameters appear to be of importance in deciding the answer to that question, four of them relating to the prehistoric events, and the remaining three to the character of obsidian hydration.

(1) What is the age of the event? The eruption is estimated to have occurred 1,117 years ago. The chronological resolution of obsidian hydration chronologies can be expected to decrease with increasing age. The appended computer simulation program, HIATUS.BAS, permits this parameter to be varied.

(2) What was the length of the cultural hiatus? This is one of the variables to be explored by the simulation discussed below. Long hiatuses are more easily detected than short ones.

(3) What was the length of the occupation period which bracketed the hiatus and from which archaeological samples of obsidian are to be drawn? For the simulation, it will be arbitrarily assumed that the samples represent activity during a 500-year period between A.D. 750 and 1250. The longer this interval, for a given hydration sample size, the poorer is the resolution with respect to a hiatus. Resolution also decreases if the bracketing occupation period is markedly asymmetrical with respect to the hiatus. HIATUS.BAS permits this parameter to be varied.

(4) Was the obsidian waste production which is represented in the sample essentially constant during the bracketing period, or highly irregular? For the simulation, it will be assumed that waste production was constant and that samples from it are essentially independent of one another. Highly irregular production or clustered samples make it more difficult to recognize a hiatus.

(5) What is the rate for the hydration of the obsidian? For the simulation, Hall's (1984) formula for Casa Diablo obsidian will be used: T = 129.7 d 1.826. Fast hydration rates give higher chronological resolution than slow ones. Linear rates give poorer chronological resolution for recent events and better resolution for older events than higherpower rates. HIATUS.BAS permits this parameter to be varied.

(6) What are the general character and magnitude of the net error in hydration readings as estimates of age? Significant components of net error probably include reading error, within-site variation in temperature history, within-source chemical variability, and perhaps also within-site variability in humidity history and a certain amount of inherent randomness in the physical process of hydration. For the simulation, the net error will be taken to be linear with respect to hydration thickness and to be normally distributed. The magnitude of the net error is a second variable to be explored by the model. A large net error makes it more difficult to detect a hiatus.

(7) What is the size of the obsidian hydration sample? This is the third variable to be explored by the model. Sample size is limited by the amount of suitable material available archaeologically and by the cost of hydration measurements. The larger the sample, the higher is the potential resolution.

THE METHOD

To simulate the distribution of obsidian hydration readings which might be found archaeologically (Figure 1), a computer program was used. For each "specimen," a date within the bracketing period (excluding the hiatus) was randomly selected. The date was converted to its hydration rim equivalent. Finally, a value from a normal distribution was randomly chosen, multiplied by the net error, and added to or subtracted from the hydration reading, which was rounded to the nearest tenth of a micron. This procedure was repeated for each of the simulated specimens in the sample, and 200 simulated frequencies were generated for each set of parameter values.

To evaluate the visibility of the hiatus in a simulated frequency of hydration readings, the dip in hydration readings corresponding to the hiatus was compared to the dips in corresponding simulations with no hiatus. Dip magnitude was defined as the sum of differences between each of the frequencies within a low in the frequency distribution and the lower of the two bracketing highs. A hiatus was considered to be visible if the dip in readings corresponding to the hiatus in the majority of simulations for a given set of variable values was greater than any dip in 190 out of 200 simulated frequencies with no hiatus.

It is also possible to make a crude estimate of the length of the hiatus, if the value for the net hydration error is taken as given. For instance, the simulated hydration reading frequency distribution shown in Figure 1 was generated for a sample size of 1,000 readings, a net error of _0.15 µm, and a hiatus length of 150 years. The dip corresponding to the hiatus is 43 readings. This dip corresponds approximately to the mean dip of 42 readings for a hiatus of 132 years in repeated simulations. The dip is greater than 90% of the dips for simulations of a hiatus of 98 years, and less than 90% of the dips for simulations of a 161-year hiatus. The best estimate of the hiatus length would therefore be 132 years, and there would be 95% confidence that the actual hiatus length was between about 98 and 161 years, under the stipulated conditions.

RESULTS AND CONCLUSIONS

The approximate thresholds of hiatus length which should be detectable with a given hydration sample size and net hydration error, under the specified conditions, are summarized in Table 1. The simulation suggests that a hiatus of as much as a century is not likely to be detectable in the archaeological record unless the hydration sample is very large, and unless the optimistic assumption is made that net error is only _0.1µm.

In addition to the results relating to the particular case of the A.D 880 eruption, three general conclusions are suggested by the simulation. First, substantial hiatuses will often be very difficult to detect in the archaeological record. Accordingly, claims for unbroken continuity in prehistoric occupations should be regarded with considerable skepticism. Second, the problem of the magnitude of net error in hydration dates merits more focused research attention than it has received. A better understanding of this variable is important not only in the hiatus problem but in other chronological uses of obsidian hydration. Third, computer simulations are potentially useful tools, both for formulating realistic research designs and for evaluating apparent patterns in actual archaeological data.

REFERENCES CITED

Hall, Matthew C.
1983 *Late Holocene Hunters-Gatherers and Volcanism in the Long Valley-Mono Basin Region: Prehistoric Culture Change in the Eastem Sierra Nevada.* Ph.D. dissertation, University of California, Riverside.
1984 Obsidian, Paleoeconomy, and Volcanism in the Eastern Sierra Nevada. Paper presented at the Great Basin Anthropological Conference.

Wickstrom, Brian and Robert Jackson
1993 *Final Report, Archaeological Resources Evaluation at CA-MNO-891, CA-MNO-2678, and CA-MNO-2679, Mono County, California.* BioSystems Analysis, Santa Cruz, California.

Wood, S. H.
1977 Distribution, Correlation, and Radiocarbon Dating of Late Holocene Tephra, Mono and Inyo Craters, Eastern California. *Geological Society of America Bulletin* 88:89-95.

Wood, S. H., and R. Brooks
1979 Panum Crater Dated 640 _ 40 yrs. B.P., Mono Craters, California. *Geological Society of America Abstract with Programs* 11:543.

APPENDIX

HIATUS.BAS is a QBasic program which simulates the effects of a hiatus on a suite of obsidian hydration measurements and evaluates the visibility of those effects against random variation.

```
DECLARE FUNCTION CalcDip% (Freq%)
DECLARE SUB CalcCutoff (Percnt%)
DECLARE SUB RunSamples (Length)
DECLARE FUNCTION HydValue% (Year!, Const1!, Const2!, Const3!)
DECLARE SUB InputParameters ()
DIM SHARED HiatusLength, Runs%, TotalDip%, FirstYear, LastYear, HiatusStart
DIM SHARED SampleSize%, StandardError, Const1, Const2, Const3, Year
DIM SHARED Freq%(200), Norm(100), Dip%(2000)
DATA 2.33, 2.05, 1.88, 1.75, 1.65, 1.56, 1.48, 1.41, 1.34, 1.28, 1.23, 1.18, 1.13, 1.08, 1.04, .99
DATA .95, .92, .88, .84, .81, .77, .74, .71, .67, .64, .61, .58, .55, .52, .5, .47, .44, .41, .39, .36,
DATA .33, .31, .28, .25, .25, .23, .2, .18, .15, .13, .1, .08, .05, .03, 0
FOR I% = 1 TO 50: READ Norm(I%): Norm(100 - I%) = -Norm(I%): NEXT I%
InputParameters
CLS : PRINT : PRINT : PRINT : PRINT : PRINT TAB(25) "(Please wait.)"
RunSamples HiatusLength
CLS : PRINT "The mean dip is"; (TotalDip% / Runs%); "."
PRINT : PRINT TAB(25) "(Please wait.)": PRINT: PRINT
RunSamples 0
CalcCutoff 1: CalcCutoff 2: CalcCutoff 5: CalcCutoff 10
END 'HIATUS.BAS

SUB CalcCutoff (Percnt%)
  DIM DipArray%(Runs%)
  FOR I% = 1 TO Runs%: DipArray%(I%) = Dip%(I%): NEXT I%
  Cutoff% = INT(Runs% * Percnt% * .01) + 1
  FOR N% = 1 TO Cutoff%
    LargeDip% = 0
    FOR P% = 1 TO Runs%
      IF DipArray%(P%) > LargeDip% THEN
        LargeDip% = DipArray%(P%): RunNr% = P%
      END IF
    NEXT P%
    IF N% < Cutoff% THEN DipArray%(RunNr%) = 0
  NEXT N%
  PRINT "Dips larger than"; LargeDip%; "are expected in no more";
  PRINT TAB(5) "than"; Percnt%; "% of samples with no hiatus."
END SUB 'CalcCutoff

FUNCTION CalcDip% (Freq%)
    DipSize% = 0: K% = 1: TestDip% = 0
  100  IF Freq%(K%) > Freq%(K% + 1) THEN 200
    K% = K% + 1: IF K% < 200 THEN 100 ELSE 900
  200  Peak1% = K%: K% = K% + 1
  300  IF Freq%(K%) < Freq%(K% + 1) THEN 400
    K% = K% + 1: IF K% < 200 THEN 300 ELSE 900
  400  K% = K% + 1
    DO UNTIL Freq%(K%) > Freq%(K% + 1): K% = K% + 1: LOOP
  500  Peak2% = K%
```

```
      IF Freq%(Peak2%) < Freq%(Peak1%) THEN 600
      LowPeak% = Peak1%: GOTO 800
   600  K% = K% + 1
      IF K% >= 200 THEN 700
      IF Freq%(K%) > Freq%(Peak2%) THEN 500 ELSE 600
   700  LowPeak% = Peak2%
   800  FOR L% = (Peak1% + 1) TO (Peak2% - 1)
         IF Freq%(L%) < Freq%(LowPeak%) THEN
            TestDip% = TestDip% + Freq%(LowPeak%) - Freq%(L%)
         END IF
      NEXT L%
      IF TestDip% > DipSize% THEN DipSize% = TestDip%
      K% = Peak2%: GOTO 100
   900  CalcDip% = DipSize%
END FUNCTION 'CalcDip%

FUNCTION HydValue% (Year, Const1, Const2, Const3)
   HydValue% = INT(10 * ((Year - Const3) / Const1) ^ (1 / Const2) + .5)
END FUNCTION 'HydValue%

SUB InputParameters
   CLS: INPUT "First year (B.P.) of the sampled obsidian production: ", FirstYear
   INPUT "      Last year of the sampled obsidian production: ", LastYear
   INPUT "                         Start of the hiatus: ", HiatusStart
   INPUT "                        Length of the hiatus: ", HiatusLength
   INPUT "        Number of Specimens in Hydration Sample: ", SampleSize%
   INPUT "     Number of Simulations to be Run (minimum 100): ", Runs%
   INPUT " Standard Error in Hydration Readings (in microns): ", StandardError
   PRINT : PRINT "The simulation uses a hydration rate of the general class"
   PRINT " Age = (a * (Hydration ^ b)) + c": PRINT
   INPUT "                  Value for a (e.g. 130, etc.): ", Const1
   INPUT "                Value for b (e.g. 1, 1.83, etc.): ", Const2
   INPUT "               Value for c (e.g., 0, -600, etc.): ", Const3
END SUB 'InputParameters

SUB RunSamples (Length)
   TotalDip% = 0
   FOR H% = 1 TO Runs%
      FOR I% = 1 TO 200: Freq%(I%) = 0: NEXT I%
      FOR J% = 1 TO SampleSize%
         DO
            Year = LastYear + (RND * (FirstYear - LastYear))
         LOOP UNTIL Year > HiatusStart OR Year < (HiatusStart - Length)
         RandomError = StandardError * Norm(INT(RND * 100)) * 10
         Rind% = HydValue%(Year, Const1, Const2, Const3) + RandomError
         IF Rind% < 0 THEN Rind% = 1
         Freq%(Rind%) = Freq%(Rind%) + 1
      NEXT J%
      Dip%(H%) = CalcDip%(Freq%)
      TotalDip% = TotalDip% + Dip%(H%)
   NEXT H%
END SUB 'RunSamples
```

Table 1. Approximate Thresholds of Hiatus Visibility

	50 samples	100 samples	200 samples	500 samples	1000 samples
_0.10 µm	180 years	140 years	120 years	90 years	80 years
_0.15 µm	220 years	200 years	170 years	140 years	130 years
_0.20 µm			200 years	180 years	170 years
_0.25 µm		>250 years		220 years	210 years
_0.30 µm					



Figure 1: Simulated frequency of obsidian hydration readings, for a 150-year hiatus, 1,000-reading sample, and cumulative error of $\pm 0.15\ \mu m$.